Using a Picture Format to Create Visit Windows
Richann Watson, DataRich Consulting

ABSTRACT

Creating visit windows is sometimes required for analysis of data. We need to make sure that we get the visit/day in the proper window so that the data can be analyzed properly. However, defining these visit windows can be quite cumbersome especially if they have to be defined in numerous programs. This task can be made easier by applying a picture format, which can save a lot of time and coding. A format is easier to maintain than a bunch of individual programs. If a change to the algorithm is required, the format can be updated instead of updating all of the individual programs containing the visit definition code.

INTRODUCTION

Hypothetically, subjects are scheduled to come in for the first 6 days and then every week thereafter for the next four weeks, allowing +/- 3 days for each week visit. In order to see if a subject deviates from the protocol schedule, the actual week is calculated based on the study days and then compared to the expected week, with unscheduled visits allowed and designated by 99.XX, where ‘99’ is used to represent unscheduled and ‘.XX’ is used to represent the sequence number, starting with 01.

THE DATA

For purposes of this paper, a dummy data set is created that contains a set of days using the DO statement with a list of items for the value of start where each item in the list uses the start TO stop syntax. Note that the DO statement can either have either all numeric or all character constants, with character constants enclosed in quotation marks. In addition, the DO statement can consist of variables. If TO or BY is used, then the start and stop must be numeric. The DO loop is executed once for each value in the list.

CREATE THE DUMMY DATA SET

To create a dummy data set that contains 38 records, the following code can be used.

```sas
data dummy;
  do DAYS = 1 to 28 by 1, 99.01 to 99.10 by .01;
    output;
  end;
run;
```
CREATE THE VISIT WINDOWS

To create the visit windows, a series of IF-THEN-ELSE statements can be incorporated into each program that requires the visit windows, such as:

```sas
data viswin_if;
  set dummy;
  length VISIT $50.;

  if DAYS = 1 then VISIT = 'Screening';

  else if DAYS = 2 then VISIT = 'Baseline';
  else if 3 <= DAYS < 7 then VISIT = catx(" ", 'Day', put(DAYS, 3.));

  else if 7 <= DAYS < 99 then do;
    weeks = round(DAYS * .142857, 1);
    visit = catx(" ", 'Week', put(weeks, 3.));
  end;

  else if DAYS >= 99 then do;
    unsch = (DAYS - 99) * 100;
    visit = catx(" ", 'Unscheduled Visit', put(unsch, z2.));
  end;

  drop weeks unsch;
run;
```

Display 1. Dummy Data
To maintain this code for every program can be quite tedious. If a change to the algorithm is required, then every affected program needs to be updated accordingly.

A simple solution is to create a format that contains this algorithm. This can be created using the ‘PICTURE’ statement instead of ‘VALUE’:

```sas
libname library "location of format catalog";

proc format lib=library;
    picture visitwin (round)
        1 = 'Screening' (noedit)
        2 = 'Baseline' (noedit)
        3-6 = 9 (prefix='Day ')
        7-98 = 09 (prefix='Week ' mult=.142857)
        99.01 - 99.99 = 00.99 (prefix='Unscheduled Visit ');
    run;
```

Based on the series of IF-THEN-ELSE statements above, days 1 and 2 have a set value. A preset value that does not require any editing can be defined in the PICTURE format using the NOEDIT option. If a value is to be preceded with a specific label for a range of values, then PREFIX= allows the user to define what label they want to use. In addition, the MULT= option allows the user to specify a multiplier that will be used to convert the existing value. In this example, the MULT=.142857 allows for a visit window of +/- 3 days, so the week will be determined based on the number of days (i.e., days will be multiplied by the multiplier prior to formatting). The ROUND option allows the visit windows to be rounded to the nearest week; otherwise, the value generated by the multiplier will be truncated (i.e., decimal portion dropped) and the integer portion will be used to print the specified format.

```sas
data viswin_pic;
    set dummy;
    length VISIT $50.;
    VISIT = put(days, visitwin.);
run;
```
Display 3. Dummy Data with Visit Window Using Picture Format

Note that the data is not quite in the desired format, but that can be fixed by using the TRANWRD, TRIM and LEFT functions.

```plaintext
data viswin_pic;
  set dummy;
  length VISIT $50.;
  VISIT = trim(left( tranwrd( put(days, visitwin.), ' 99.', '') ) );
run;
```

Display 4. Final Dummy Data with Visit Window Using Picture Format

An alternative approach to getting the Unscheduled visits to only keep the decimal portion of days is to use the MULT= option on the 'Unscheduled Visit' portion of the picture format.

```plaintext
proc format lib=library;
  picture visitwin (round)
    1 = 'Screening' (noedit)
    2 = 'Baseline' (noedit)
    3-6 = 9 (prefix='Day ')
    7-98 = 09 (prefix='Week ' mult=.142857)
    99.01 - 99.99 = 00.99 (prefix='Unscheduled Visit' mult=10000);
run;
```
The only drawback with this is that the ‘Unscheduled Visit’ will still have a decimal form as illustrated in Display 5.

<table>
<thead>
<tr>
<th>DAYS</th>
<th>VISIT</th>
</tr>
</thead>
<tbody>
<tr>
<td>99.01</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.02</td>
<td>1.00</td>
</tr>
<tr>
<td>99.03</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.04</td>
<td>2.00</td>
</tr>
<tr>
<td>99.05</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.06</td>
<td>3.00</td>
</tr>
<tr>
<td>99.07</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.08</td>
<td>4.00</td>
</tr>
<tr>
<td>99.09</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.1</td>
<td>5.00</td>
</tr>
<tr>
<td>99.01</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.02</td>
<td>6.00</td>
</tr>
<tr>
<td>99.03</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.04</td>
<td>7.00</td>
</tr>
<tr>
<td>99.05</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.06</td>
<td>8.00</td>
</tr>
<tr>
<td>99.07</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.08</td>
<td>9.00</td>
</tr>
<tr>
<td>99.09</td>
<td>Unscheduled Visit</td>
</tr>
<tr>
<td>99.1</td>
<td>10.00</td>
</tr>
</tbody>
</table>

Display 5. Unscheduled Visits with MULT = 10000 option

CONCLUSION

It is sometimes said a picture is worth a thousand words. In this case it may not be a thousand words, but it is definitely worth something. For projects that require a consistent visit designation across all data sets or if the actual visit occurred according to a pre-defined schedule creating a PICTURE FORMAT to determine these visit windows can save programming time and lots of headaches especially if more visits are added or if the visit windows change.

REFERENCES


ACKNOWLEDGMENTS

Thanks to my former colleagues at inVentiv Health, who reviewed this paper and supported me while I worked through this process. In addition, I would like to thank Rick Langston for answering my question regarding the NOEDIT option.

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. © indicates USA registration.

Other brand and product names are trademarks of their respective companies.